

Task

Implement a Caesar cipher, both encoding and decoding. 

The key is an integer from 1 to 25.

This cipher rotates (either towards left or right) the letters of the alphabet (A to Z).

The encoding replaces each letter with the 1st to 25th next letter in the alphabet (wrapping Z to A).

So key 2 encrypts "HI" to "JK", but key 20 encrypts "HI" to "BC".

This simple "mono-alphabetic substitution cipher" provides almost no security, because an attacker who has the encoded message can either use frequency analysis to guess the key, or just try all 25 keys.

Caesar cipher is identical to Vigenère cipher with a key of length 1. 

Also, Rot-13 is identical to Caesar cipher with key 13.



Related tasks

 Rot-13
 Substitution Cipher
 Vigenère Cipher/Cryptanalysis


beads 1 program 'Caesar cipher'
calc main_init
	var str = "The five boxing wizards (🤖) jump quickly."
	log "Plain: {str}"
	str = Encrypt(str, 3)
	log "Encrypted: {str}"
	str = Decrypt(str, 3)
	log "Decrypted: {str}"
 
//  encrypt a string by shifting the letters over by a number of slots
//  pass through any characters that are not in the Roman alphabet
calc Encrypt(
	input:str  --- string to encrypt
	nshift	 --- number of characters to slide over
	) : str	--- encrypted string
 
	var newStr = ""
	loop from:1 to:str_len(input) count:myCount
		var unicode : num = from_char(subset(input, from:myCount, len:1))
		if unicode >= 65 and unicode <= 90
			unicode = mod((unicode - 65 + nshift), 26) + 65
		elif unicode >= 97 and unicode <= 122
			unicode = mod((unicode - 97 + nshift), 26) + 97
		newStr = newStr & to_char(unicode)
	return newStr
 
//   undo the encryption 
calc Decrypt(
	input:str 
	nshift
	) : str
	//  we could also just shift by 26-nshift, same as going in reverse
	return Encrypt(input, -nshift)